
#define DEBUG true
#define KEYPAD A5
#define INTERVAL 300
#define FIX 3  //the number of displayable decimal places for floats
uint16_t thresholds[16] = { 55, 58, 62, 66, 74, 80, 88, 96, 115, 130, 150, 177, 255, 340, 512, 1023 };
char keys[] = { '+', 'E', '.', '0', '-', '3', '2', '1', '*', '6', '5', '4', '/', '9', '8', '7' };
char operand[12];  //current operand buffer  (12 chars is reasonable)
char newKey[2];    //buffer for conversion of char to string

void setup() {
  if (DEBUG) {
    Serial.begin(9600);
    while (!Serial) {};
  }
}

void loop() {
  char ch = getEntry();  //Acquire the keypad character
  process(ch);           //Do one of three things(operand/operator/ENTER)
}

void process(char ch) {
  if (isDigit(ch) || ch == '.') {  // Is the character a digit?
    strncat(operand, &ch, 1);      // If so add char to the end operand 
    if (DEBUG) {
      Serial.print("Digit\t");
      Serial.println(operand);
    }
  } else if (isOperator(ch)) {  //Is it an operator?
    if (DEBUG) {                //If so, just display it for now...
      Serial.println("Operator");
    }
  } else {                         //OK, it must be the (E)NTER key
    float result = atof(operand);  //convert string to its float equivalent
    if (DEBUG) {
      Serial.print("ENTER\t");
      Serial.println(result, FIX);  
    }
    operand[0] = '\0';  //reuse the operand buffer for the next number
  }
}



[bookmark: _GoBack]// 'predicate' function determines if input is one of four operators...
boolean isOperator(char ch) {
  return ch == '+' || ch == '-' || ch == '*' || ch == '/';
};

//reads an analog voltage from the keypad and maps it to a key
char getEntry() {
  uint16_t reading;
  while ((reading = analogRead(KEYPAD)) == 0) {};

  //loop through the thresholds array looking for a match for reading...
  uint8_t index = 0;
  while (reading > thresholds[index]) {
    index++;
  }
  //publish...
  if (DEBUG) {
    Serial.print(reading);
    Serial.print('\t');
    Serial.print(keys[index]);
    Serial.print('\t');
  }
  delay(INTERVAL);  //pause...
  return keys[index];
}


